********* ============================================================= ********
*** Replication Material for: ***
*** Changing the Lens: The contingency of results from conjoint experiments on 
*** the outcome variable and the estimand, Research & Politics ***
*** Author: Clareta Treger, PhD

*** Updated: April 1, 2025 *** 
*** Finalized: May 13, 2025 ***
********* ============================================================= ********


********* ============================================================= ********
							*** /// Original Israeli Study \\\ ***
********* ============================================================= ********

*** === Data File === ***

use "Israel Data_Changing the Lens.dta"

*** === Sample Demographics === ***

* Note: DROP variable indicated low quality responses that were dropped from the analysis 
tab1 gender age edu origin ussr relig region if DROP !=1 


*** === AMCEs Analysis - Table C1 in the SM, Columns (1)-(3) === ***

* Rescaling the rating outcome to 0-1	
* The formula for the rescaling is: X= (x-a)/(b-a), where a is the min and b is the max in the original scale. 
recode support_ (1=0) (2=0.167) (3=0.333) (4=0.5) (5=0.667) (6=0.833) (7=1), gen(support_res)

* DV: Rating Continuous
reg support_res ib2.coer i.public_support ib2.effect ib5.promoter ib2.imp if DROP !=1, vce(cluster rid) baselevel 
eststo il_amce_c_res

*DV: Rating Binary 
*Recoding the rating DV to a binary DV as in the original study (Treger 2023)
recode support_ (1/4=0) (5/7=1), gen(supported)

reg supported ib2.coer i.public_support ib2.effect ib5.promoter ib2.imp if DROP !=1, vce(cluster rid) baselevel 
eststo il_amce_b

*DV: Forced Choice ***

reg choice_ ib2.coer i.public_support ib2.effect ib5.promoter ib2.imp if DROP !=1, vce(cluster rid) baselevel 
eststo il_amce_choice

*** === MMs Analysis - Table C2 in the SM, Columns (1)-(3) === ***

*DV: Rating Continuous 
reg support_res i.coer i.public_support i.effect i.promoter i.imp if DROP !=1, vce(cluster rid) baselevel
margins coer public_support effect promoter imp if DROP !=1, post
eststo il_mm_res

*DV: Rating Binary 
reg supported i.coer i.public_support i.effect i.promoter i.imp if DROP !=1, vce(cluster rid) baselevel
margins coer public_support effect promoter imp if DROP !=1, post
eststo il_mm_b

*DV: Forced Choice 
reg choice_ i.coer i.public_support i.effect i.promoter i.imp if DROP !=1, vce(cluster rid) baselevel
margins coer public_support effect promoter imp if DROP !=1, post
eststo il_mm_choice

*** === Figure 3, panels (A) and (B) === ***
	
*AMCEs - Rating continuous + Froced Choice (Panel A)
coefplot  /// 
(il_amce_c_res, label("Rating outcome") msymbol(circle) mcolor(black) mfcolor(black) ciopts(lcolor(black) lwidth(thin))) /// 
(il_amce_choice, label("Forced-choice outcome") msymbol(triangle) mcolor(blue) mfcolor(blue) ciopts(lcolor(blue) lwidth(thin))), /// 
mlabsize(medium) /// 
headings(1.coer="{bf: Coercion level}" 1.public_support="{bf: Public support}" 2.effect="{bf:Effectiveness}" 1.promoter="{bf:Promoter}" /// 
2.imp="{bf:Implementation}", ///
labcolor(black) labgap (3) labsize (small)) mlabsize(small) msize(small) lwidth(thin) /// 
coeflabels(2.coer="Nudge" 1.effective_="Somewhat effective" ///
2.effective_="Effective" 3.effective_="Very effective" 4.effective_="Unknown" 4.promoter="NGO" , labsize(small)) ///
order (1.coer 2.coer 3.coer 4.coer 1.public_support 2.public_support 3.public_support 4.public_support 5.public_support ) ///
ciopts(lcolor(black) mcolor(black)) ///
drop (2.effect 1.effect 4.effect 3.effect 1.promoter 3.promoter 6.promoter 7.promoter 2.promoter 4.promoter 5.promoter 2.imp 1.imp _cons) /// 
xline(0, lcolor(black)) xscale(range(-.2 (.1) .2)) base ///
xtitle("AMCE: Effect on probability of Support", size(small)) scheme (s1color) mcolor(black) msymbol(circle) ///
title ("(A) Israel")

*MMs - Rating continuous + Froced Choice (Panel B) === ***
coefplot ///
(il_mm_res, label("Rating outcome") msymbol(circle) mcolor(black) mfcolor(black) ciopts(lcolor(black) lwidth(thin))) /// 
(il_mm_choice, label("Forced-choice outcome") msymbol(triangle) mcolor(blue) mfcolor(black) ciopts(lcolor(blue) lwidth(thin))), ///  
mlabsize(medium) /// 
headings(1.coer="{bf: Coercion level}" 1.public_support="{bf: Public support}" 2.effect="{bf:Effectiveness}" 1.promoter="{bf:Promoter}" /// 
2.imp="{bf:Implementation}", ///
labcolor(black) labgap (3) labsize (small)) mlabsize(small) msize(small) lwidth(thin) /// 
coeflabels(2.coer="Nudge" 1.effective_="Somewhat effective" ///
2.effective_="Effective" 3.effective_="Very effective" 4.effective_="Unknown" 4.promoter="NGO" , labsize(small)) ///
order (1.coer 2.coer 3.coer 4.coer 1.public_support 2.public_support 3.public_support 4.public_support 5.public_support ) ///
ciopts(lcolor(black) mcolor(black)) ///
drop (2.effect 1.effect 4.effect 3.effect 1.promoter 3.promoter 6.promoter 7.promoter 2.promoter 4.promoter 5.promoter 2.imp 1.imp _cons) ///
xline(0.5, lcolor(black)) xscale(range(0.4 0.7)) base ///
xtitle("Marginal means", size(small)) scheme (s1color) mcolor(black) msymbol(circle) ///
title ("(B) Israel")

*** === SM Figure C1, Israeli panel === ***
coefplot  /// 
(il_amce_c_res, label("Rating outcome") msymbol(circle) mcolor(black) mfcolor(black) ciopts(lcolor(black) lwidth(thin))) /// 
(il_amce_choice, label("Forced-choice outcome") msymbol(triangle) mcolor(blue) mfcolor(black) ciopts(lcolor(blue) lwidth(thin))), /// 
mlabsize(small) /// 
headings(1.coer="{bf: Coercion level}" 1.public_support="{bf: Public support}" 2.effect="{bf:Effectiveness}" 1.promoter="{bf:Promoter}" /// 
2.imp="{bf:Implementation}", ///
labcolor(black) labgap (3) labsize (vsmall)) mlabsize(tiny) msize(vsmall) lwidth(thin) /// 
coeflabels(2.coer="Nudge" 1.effective_="Somewhat effective" ///
2.effective_="Effective" 3.effective_="Very effective" 4.effective_="Unknown" 4.promoter="NGO" , labsize(vsmall)) ///
order (1.coer 2.coer 3.coer 4.coer 1.public_support 2.public_support ///
3.public_support 4.public_support 5.public_support 2.effect 1.effect 4.effect 3.effect /// 
1.promoter 3.promoter 6.promoter 7.promoter 2.promoter 4.promoter 5.promoter 2.imp 1.imp) ///
ciopts(lcolor(black) mcolor(black)) drop (_cons) xline(0) xscale(range(-.2 .2)) base ///
xtitle("AMCE: Effect on probability of Support", size(small)) scheme (s1color) mcolor(black) msymbol(circle) ///
title ("Israel")

*** === SM Figure C2, Israeli panel === ***
coefplot ///
(il_mm_res, label("Rating outcome") msymbol(circle) mcolor(black) mfcolor(black) ciopts(lcolor(black) lwidth(thin))) /// 
(il_mm_choice, label("Forced-choice outcome") msymbol(triangle) mcolor(blue) mfcolor(black) ciopts(lcolor(blue) lwidth(thin))), ///  
mlabsize(small) /// 
headings(1.coer="{bf: Coercion level}" 1.public_support="{bf: Public support}" 2.effect="{bf:Effectiveness}" 1.promoter="{bf:Promoter}" /// 
2.imp="{bf:Implementation}", ///
labcolor(black) labgap (3) labsize (vsmall)) mlabsize(tiny) msize(vsmall) lwidth(thin) /// 
coeflabels(2.coer="Nudge" 1.effective_="Somewhat effective" ///
2.effective_="Effective" 3.effective_="Very effective" 4.effective_="Unknown" 4.promoter="NGO" , labsize(vsmall)) ///
order (1.coer 2.coer 3.coer 4.coer 1.public_support 2.public_support ///
3.public_support 4.public_support 5.public_support 2.effect 1.effect 4.effect 3.effect /// 
1.promoter 3.promoter 6.promoter 7.promoter 2.promoter 4.promoter 5.promoter 2.imp 1.imp) ///
ciopts(lcolor(black) mcolor(black)) drop (_cons) xline(0.5, lcolor(black)) xscale(range(0.4 0.7)) base ///
xtitle("Marginal means", size(small)) scheme (s1color) mcolor(black) msymbol(circle) ///
title ("Israel")

********* ============================================================= ********
			*** /// Replication of U.S. Study (Treger 2023) \\\ ***
********* ============================================================= ********

*** === Data File === ***

use "US Data_Changing the Lens.dta"

* Note: LQ  variable indicated low quality responses that were dropped from the analysis
 
*** === AMCEs Analysis - Table C1 in the SM, Columns (4)-(6) === ***

* Rescaling the rating outcome to 0-1	
* The formula for the rescaling is: X= (x-a)/(b-a), where a is the min and b is the max in the original scale. 
recode support_ (1=0) (2=0.167) (3=0.333) (4=0.5) (5=0.667) (6=0.833) (7=1), gen(support_res)


* DV: Rating Continuous

reg support_res i.coer i.public_support i.effective_ i.promoter i.imp if LQ !=1, vce(cluster rid) baselevel
eststo us_amce_c_res


*DV: Rating Binary 
*Recoding the rating DV to a binary DV as in the original study (Treger 2023)***

recode support_ (1/4=0) (5/7=1), gen(supported)

reg supported i.coer i.public_support i.effective_ i.promoter i.imp if LQ !=1, vce(cluster rid) baselevel
eststo us_amce_b

***DV: Forced Choice ***

reg choice_ i.coer i.public_support i.effective_ i.promoter i.imp if LQ !=1, vce(cluster rid) baselevel
eststo us_amce_choice


*** === MMs Analysis - Table C2 in the SM, Columns (4)-(6) === ***

*DV: Rating Continuous *

reg support_res i.coer i.public_support i.effective_ i.promoter i.imp if LQ !=1, vce(cluster rid) baselevel
margins coer public_support effective_ promoter imp if LQ !=1, post
eststo us_mm_res


*DV: Rating Binary  *
reg supported i.coer i.public_support i.effective_ i.promoter i.imp if LQ !=1, vce(cluster rid) baselevel
margins coer public_support effective_ promoter imp if LQ !=1, post
eststo us_mm_b

***DV: Forced Choice
reg choice_ i.coer i.public_support i.effective_ i.promoter i.imp if LQ !=1, vce(cluster rid) baselevel
margins coer public_support effective_ promoter imp if LQ !=1, post
eststo us_mm_choice

*** === Figure 3, panels (C) and (D) === ***
	
*AMCEs - Rating continuous + Froced Choice (Panel C)
coefplot ///
(us_amce_c_res, label("Rating outcome") msymbol(circle) mcolor(black) mfcolor(black) ciopts(lcolor(black) lwidth(thin))) /// 
(us_amce_choice, label("Forced-choice outcome") msymbol(triangle) mcolor(blue) mfcolor(blue) ciopts(lcolor(blue) lwidth(thin))), /// 
mlabsize(medium) /// 
headings(1.coer="{bf: Coercion level}" 1.public_support="{bf: Public support}" 1.effective_="{bf:Effectiveness}" 1.promoter="{bf:Promoter}" /// 
2.imp="{bf:Implementation}", ///
labcolor(black) labgap (3) labsize (small)) mlabsize(small) msize(small) lwidth(thin) /// 
coeflabels(2.coer="Nudge" 1.effective_="Somewhat effective" ///
2.effective_="Effective" 3.effective_="Very effective" 4.effective_="Unknown" 4.promoter="NGO" , labsize(small)) ///
order (1.coer 2.coer 3.coer 4.coer 1.public_support 2.public_support ///
3.public_support 4.public_support 5.public_support 1.effective_ 2.effective_ 3.effective_ 4.effective_ /// 
1.promoter 6.promoter 2.promoter 3.promoter 4.promoter 5.promoter  2.imp 1.imp) ///
ciopts(lcolor(black)) ///
drop (1.effective_ 2.effective_ 3.effective_ 4.effective_ 1.promoter 6.promoter 2.promoter 3.promoter 4.promoter 5.promoter  2.imp 1.imp _cons) ///
xline(0, lcolor(black))  xscale(range(-0.2 0.2)) base ///
xtitle("AMCE: Effect on probability of support", size(small)) scheme (s1color) ///
title ("(C) United States")


*MMs - Rating continuous + Froced Choice (Panel D)
	
coefplot /// 
(us_mm_res, label("Rating outcome") msymbol(circle) mcolor(black) mfcolor(black) ciopts(lcolor(black) lwidth(thin))) /// 
(us_mm_choice, label("Forced-choice outcome") msymbol(triangle) mcolor(blue) mfcolor(blue) ciopts(lcolor(blue) lwidth(thin))), /// 
mlabsize(medium) /// 
headings(1.coer="{bf: Coercion level}" 1.public_support="{bf: Public support}" 2.effect="{bf:Effectiveness}" 1.promoter="{bf:Promoter}" /// 
2.imp="{bf:Implementation}", ///
labcolor(black) labgap (3) labsize (small)) mlabsize(small) msize(small) lwidth(thin) /// 
coeflabels(1.coer="Information" 2.coer="Nudge" 1.effective_="Somewhat effective" ///
2.effective_="Effective" 3.effective_="Very effective" 4.effective_="Unknown" 4.promoter="NGO" , labsize(small)) ///
order (1.coer 2.coer 3.coer 4.coer 1.public_support 2.public_support ///
3.public_support 4.public_support 5.public_support 2.effect 1.effect 4.effect 3.effect /// 
1.promoter 6.promoter 3.promoter  2.promoter 4.promoter 5.promoter 2.imp 1.imp) ///
ciopts(lcolor(black) mcolor(black)) ///
drop (1.effective_ 2.effective_ 3.effective_ 4.effective_ 1.promoter 6.promoter 2.promoter 3.promoter 4.promoter 5.promoter  2.imp 1.imp _cons) ///
xline(0.5, lcolor(black)) xscale(range(0.4 (0.1) 0.7)) base ///
xtitle("Marginal means", size(small)) scheme (s1color) mcolor(black) msymbol(circle) ///
title ("(D) United States")

*** === SM Figure C1, US panel === ***

coefplot ///
(us_amce_c_res, label("Rating outcome") msymbol(circle) mcolor(black) mfcolor(black) ciopts(lcolor(black) lwidth(thin))) /// 
(us_amce_choice, label("Forced-choice outcome") msymbol(triangle) mcolor(blue) mfcolor(black) ciopts(lcolor(blue) lwidth(thin))), /// 
mlabsize(small) /// 
headings(1.coer="{bf: Coercion level}" 1.public_support="{bf: Public support}" 1.effective_="{bf:Effectiveness}" 1.promoter="{bf:Promoter}" /// 
2.imp="{bf:Implementation}", ///
labcolor(black) labgap (3) labsize (vsmall)) mlabsize(tiny) msize(vsmall) lwidth(thin) /// 
coeflabels(2.coer="Nudge" 1.effective_="Somewhat effective" ///
2.effective_="Effective" 3.effective_="Very effective" 4.effective_="Unknown" 4.promoter="NGO" , labsize(vsmall)) ///
order (1.coer 2.coer 3.coer 4.coer 1.public_support 2.public_support ///
3.public_support 4.public_support 5.public_support 1.effective_ 2.effective_ 3.effective_ 4.effective_ /// 
1.promoter 6.promoter 2.promoter 3.promoter 4.promoter 5.promoter  2.imp 1.imp) ///
ciopts(lcolor(black)) drop (_cons) xline(0) base ///
xtitle("AMCE: Effect on probability of support", size(small)) scheme (s1color) ///
title ("United States")

*** === SM Figure C2, US panel === ***

coefplot /// 
(us_mm_res, label("Rating outcome") msymbol(circle) mcolor(black) mfcolor(black) ciopts(lcolor(black) lwidth(thin))) /// 
(us_mm_choice, label("Forced-choice outcome") msymbol(triangle) mcolor(blue) mfcolor(black) ciopts(lcolor(blue) lwidth(thin))), /// 
mlabsize(small) /// 
headings(1.coer="{bf: Coercion level}" 1.public_support="{bf: Public support}" 2.effective_="{bf:Effectiveness}" 1.promoter="{bf:Promoter}" /// 
2.imp="{bf:Implementation}", ///
labcolor(black) labgap (3) labsize (vsmall)) mlabsize(tiny) msize(vsmall) lwidth(thin) /// 
coeflabels(1.coer="Information" 2.coer="Nudge" 1.effective_="Somewhat effective" ///
2.effective_="Effective" 3.effective_="Very effective" 4.effective_="Unknown" 4.promoter="NGO" , labsize(vsmall)) ///
order (1.coer 2.coer 3.coer 4.coer 1.public_support 2.public_support ///
3.public_support 4.public_support 5.public_support 2.effective_ 1.effective_ 4.effective_ 3.effective_ /// 
1.promoter 6.promoter 3.promoter  2.promoter 4.promoter 5.promoter 2.imp 1.imp) ///
ciopts(lcolor(black) mcolor(black)) drop (_cons) xline(0.5) xscale(range(0.4 (0.05) 0.7)) base ///
xtitle("Marginal means", size(small)) scheme (s1color) mcolor(black) msymbol(circle) ///
title ("United States")

*** === End of Do File === ***
